SC968 PANEL DATA METHODS for SOCIOLOGISTS

WORKSHEET FOR LECTURE 1 part 1
1.1.0
OBJECTIVES:
· To recap on basic commands in STATA

· To recap the use of OLS, Probit and Logit regressions
· To perform preliminary analyses of the relationship between mental health and gender using these methods
· To examine results critically

1. 1.1
HOW TO WORK THROUGH THESE EXERCISES

Each set of exercises contains a work sheet and a STATA do-file. 

The worksheet will explain what you need to do, and why you need to do it. Sometimes the sheet will tell you what STATA commands are appropriate – these are printed in bold type. Sometimes, the sheet will ask you to do something without telling you what STATA command to use. In this case, have a go yourself, and look at the do-file to check whether you have done the right thing. If you get stuck, you can also look at the do-file, but make sure you understand the reasons why a certain task is done in a certain way before you move on. The worksheet also asks you to fill in answers at various points. We will distribute “answer sheets” at a later stage.

1. 1.2
GETTING STARTED


Our teaching data is a subset of the British Household Panel Survey (BHPS).


You need to sign the user agreement before you download the data.


Online documentation for BHPS is at 
https://www.iser.essex.ac.uk/bhps/documentation

Set up a working directory. This could be your m: drive, a subset of your m: drive, or a memory stick. In these instructions, m:/ denotes your working directory – so if you are not going to use m:/ as your working directory, substitute the appropriate drive and path.


Download the data from the course website: https://www.iser.essex.ac.uk/resources/module-sc968/course-materials, and save it in your working directory as “teaching.dta”.

In order to download the data, you will need a password, which you can obtain after signing the user agreement.


Try to keep a “clean” copy of the data in your working directory, and save any other datafiles which you may produce under different names.


Open STATA, and set your working directory as your home directory


cd m:


Open the teaching dataset



use teaching, clear

The , clear option is not essential, but it is good practice when opening a new dataset, and makes sure you start with a clean slate.


Before you start:


If you want to learn more about any command, you can look in the STATA manuals, or type help into the command bar, followed by the command you want help with. 

Remember that STATA commands usually have a full name and an abbreviated name – e.g. “tabulate” can be shortened to “tab”

Variable names can also be abbreviated, provided they remain unique.

1.1.3
START LOOKING AT THE DATA


describe
gives you information on the variables contained in the data set

summarize
tells you about sample sizes, means, and the range of values.

Try the summarize command with the weekly hours variable jbhrs



summarize jbhrs

Question: Is the mean generated by this command very meaningful? If not, why not?
	



Try the codebook and labelbook commands to inspect the data for the way in which missing and out-of-range values are coded; you can also examine particular variables in more detail with the tabulate command.

Once you know which codes apply to missing or out-of-range values, you can fix the problem either with the mvdecode command, which declares certain values as missing 


mvdecode jbhrs, mv(-9 -8 -2 -1 99)


Or you could get the same result by using an if statement to filter out these values (we’ll see more of these later).
1.1.4
MEAN AND MEDIAN HOURS OF WORK


Task: use the summarize command to find mean and median hours of work, (a) for everyone, and (b) excluding people whose reported hours of work are zero. Remember to do something sensible with missing values! You will need to use the ,d option to get medians.
	
	Mean hours of work
	Median hours of work

	Whole sample
	
	

	Excluding zero hours of work
	
	


Check your answers, and the commands you used to get them, in the do-file.
1.1.5
START LOOKING AT MENTAL HEALTH AND GENDER


tab hlghq2 sex if hlghq2 >= 0, col nof


Check you know what every element in this command is doing

In very basic terms, what does this table tell you about the relationship between mental health and gender?
	



Generate and examine a dichotomous variable indicating risk of minor psychiatric disorder. This variable will be based on hlghq2, and will take the value 1 for those individuals whose score on hlghq2 is greater than 2.


gen PM = hlghq2



recode PM 0/2 = 0 3/12 = 1 -9 -8 -7 = .



tab hlghq2 PM

Cross-tabulate the new PM variable with sex, adding the chi2 option (try to do this yourself, look in the do-file if you get stuck).
How much more likely are women than men to be at risk of minor psychiatric disorder? Is this difference statistically significant?

	


1.1.6 
OLS REGRESSION - REGRESS LIKERT SCORE ON SEX AND AGE

We are going to run a simple OLS regression, using the continuous LIKERT scale as the dependent variable, and with sex, age and age squared as explanatory variables.


Prepare the dependent variable - call it LIKERT, and copy it from hlghq1, but with missing values removed

gen LIKERT = hlghq1

recode LIKERT -7 -8 -9 = .

tab LIKERT


No special preparation for age is needed – use the tab command to check that there are no codes which need to be omitted. Generate an age-squared variable:
tab age 



gen age2 = age * age



Check that you know why we include an age squared variable in regressions!

Generate a dichotomous variable indicating whether the respondent is female (females take value 1; males take value 0). 

Check that you know why we are coding the variable this way instead of using the “sex” variable as it appears in the data set. Tabulate the variable and check that 50.99% of your sample are females. 

We are now ready to proceed with a simple OLS regression.



reg LIKERT age age2 female


What does this add to the tabulation we did earlier?
	



The Likert score increases in age and decreases in age squared. Calculate the age at which it starts to decrease!
	



What proportion of the variation in the dependent variable is explained by this regression?

	



Can you think of ways in which we might improve the specification? (This should not be difficult, since we have started with an extremely basic specification!)
	


1.1.7
IMPROVING THE OLS SPECIFICATION


We are going to explore improving the specification in two ways: by adding additional explanatory variables, and by investigating whether the coefficients vary between different groups of people. 

First, let’s see whether the coefficients on age vary by gender. We could use separate regressions for men and women:
bysort female: reg LIKERT age age2


This shows us that the coefficients are somewhat different for men and women, but it doesn’t allow us to assess whether these differences are statistically significant.

Instead, we may add interaction terms between sex and the age variables. We can easily do this in stata by including two ## in between the variables we wish to interact. We also need to include a prefix c. to tell stata that age and age2 are continuous and a prefix i. for female to indicate a dummy/categorical variable:
reg LIKERT c.age##i.female c.age2##i.female 


Comparing this regression with the separate regressions for men and women, we see that the coefficients on age and age squared in the regression with interactions are identical to the coefficients in the “men only” regression. And the coefficients in the “women only” regression are equal to the main coefficients plus the interaction coefficients in the regression with interactions. 


Neither of the interaction terms is significant. This tells us that the effects of age on mental health are the same for both men and women. 

Let’s now add some additional variables to the specification: education, whether the person has a partner, and their work status. 

Education: create two dummy variables: a dummy ed_deg indicating whether the person has a first or higher degree, and a dummy ed_sec indicating whether the person has A levels or better (nursing, teaching, other higher), but no degree. Remember to exclude missing values – count “other qf” and “still at school no qf” as belonging to your omitted category.

Tabulate your education variables: if you don’t get the results below, check out the syntax in the do-file. 

      Total       26,177      100.00

                                                

          1        9,601       36.68      100.00

          0       16,576       63.32       63.32

                                                

     ed_sec        Freq.     Percent        Cum.

. tab ed_sec

      Total       26,177      100.00

                                                

          1        2,699       10.31      100.00

          0       23,478       89.69       89.69

                                                

     ed_deg        Freq.     Percent        Cum.

. tab ed_deg



Partnership status: generate a dummy variable called "partner" indicating whether the person lives with a spouse or partner. Check you get the following distribution.

      Total       27,700      100.00

                                                

          1       18,375       66.34      100.00

          0        9,325       33.66       33.66

                                                

    partner        Freq.     Percent        Cum.

. tab partner

(9852 missing values generated)



Work status: generate a dummy variable called “ue_sick” indicating whether the person is unemployed or sick. Again, check your results:

      Total       27,678      100.00

                                                

          1        2,177        7.87      100.00

          0       25,501       92.13       92.13

                                                

    ue_sick        Freq.     Percent        Cum.

. tab ue_sick



Physical health: generate a variable called “badhealth” which is a copy of hlstat, omitting missing values (ie, those less than zero). You should have 25934 observations, with 563 coded as 5 (very poor).

Number of children aged between 0 and 2 There is already a variable in the data, nch02. Check that this variable is “clean” and ready to use. 

We’re now ready to regress Likert scores on the original variables plus the new ones we’ve just created.

reg LIKERT age age2 female ed_deg ed_sec partner ue_sick badhealth nch02

Because there are multiple observations on individuals, try the ,cluster(pid) option

reg LIKERT age age2 female ed_deg ed_sec partner ue_sick badhealth nch02, cluster(pid)


These are much more interesting regressions! 


If you have time, test whether some of the new coefficients vary by sex – use the bysort command to look for some likely candidates, and use interaction terms to test whether these coefficients are indeed significantly different between men and women. What do you find?
	


1.1.8
LOGIT AND PROBIT REGRESSIONS


Perform logit and probit regressions using the same specification as above, but using the dichotomous PM variable which we created earlier as the dependent variable. 
logit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02
probit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02

Does Amemiya's scaling factor 1.6 * (Probit coeffs) = (Logit coeffs) work?

	



Estimate the same regressions but requesting marginal effects. Try to estimate marginal effects at the means (MEM), average marginal effects (AME) and marginal effects at representative values (MER). How do they differ? 
logit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02
margins, dydx(*) atmeans

margins, dydx(*)
margins, dydx(*) at(age=(20 30 40 50))

How do they differ?

Try the same with probit.
probit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02
When you have time, compare logit and probit marginal effects with OLS coefficients


On the logit model, test whether the coefficient on being unemployed or on sick leave is statistically different from the “badhealth” coefficient.
logit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02


test ue_sick = badhealth

STATA does not reject the hypothesis that they are identical. 

However, we shouldn’t jump to the conclusion that being in poor health has the “same effect” on mental health as being unemployed or being on sick leave. Why?
	



Using the logit estimates, run the likelihood ratio test on whether the "partner" coefficient should be dropped. (If the logit regression above was the last regression you performed, you don’t need to repeat it in the first line below.)
logit PM age age2 female ed_deg ed_sec partner ue_sick badhealth nch02
estimates store ALL
gen byte SAMPLE = e(sample)
logit PM age age2 female ed_deg ed_sec ue_sick badhealth nch02 if e(sample)
estimates store DROP_PNR
lrtest ALL DROP_PNR
drop SAMPLE


Should the “partner” variable be dropped?
	


1.1.9 
AT THE END OF THE SESSION

save your working data set to your home directory, under the name SESSION1-1.dta
save SESSION1-1
SC968 PANEL DATA METHODS for SOCIOLOGISTS

WORKSHEET FOR LECTURE 1 part 2
1.2.0
OBJECTIVES

· To learn how to use data management commands: merging, appending, converting from long to wide form
· To run simple models which use longitudinal data

1.2.1
APPENDING FILES

There are two data sets in the course directory named extra1.dta and extra2.dta. Download them to your working directory, and have a look at them.

As well as the variables hid, pid and wave (which are in the main data set) these two data sets contain a variable indicating whether the respondent smokes, and a variable indicating the year they moved to their current address.

extra1 contains these extra variables from waves 1-6 of the BHPS, and extra2 contains the extra variables from the remaining waves.
Because the two data sets relate to different observations, they should be appended rather than merged.
Try appending the files, and saving them to a new file called newfile1 in your home directory.

use extra1

append using extra2

save newfile1
Check that nothing has gone wrong, and you have the expected number of variables and observations

des using extra1

des using extra2

des using newfile1


(or just des, if you still have newfile1 as your active data set)

1.2.2
MERGING FILES

We want to merge the data in newfile1.dta with our main data set as saved at the end of last session. In order to merge the two data sets, STATA needs to know which variable(s) to use to match up the two sets of variables.

The best way to sort individual-level data in BHPS is by pid and wave
use SESSION1-1
sort pid wave

merge 1:1 pid wave using newfile1 

Check that you have merged properly, and drop the _merge variable when you are happy that the merging process has been successful.

drop _merge
Save the resulting data set in your home directory as SESSION1-2.dta
save SESSION1-2

This is an example of one-to-one merging.

For an example of one-to-many merging, use a file called hhtenure.dta taken from the BHPS household files, which as well as the hid and wave variables, contains a variable indicating housing tenure.

There is only one observation per household, and we want to link this household data to every individual in that household. This is a one-to-many operation - each observation in the using data will be linked to several observations in the master data.

Merge the two files by wave and hid

use hhtenure

merge 1:m wave hid using SESSION1-2

Once again, check that the merge was successful, then drop _merge and save the data set.

drop _merge 
save SESSION1-2, replace

1.2.3
CONVERTING BETWEEN LONG AND WIDE FORMS
Work with a subset of data for this section, so you can more easily see what's going on. Sort by pid and wave, and keep the first 50 observations, and the following variables: pid, wave, age, sex, LIKERT. 
sort pid wave

keep in 1/50

keep pid wave age sex LIKERT
Type list or list, clean  and observe that the data is in "long" form, with one observation for each person/year
Check out which variables are constant for each individual (pid and sex), which change in predictable ways (age and wave), and which are variable (LIKERT)
Reshape the data to wide form
reshape wide LIKERT age, i(pid) j(wave)

Note that we don’t include sex in the stub names, because it doesn’t vary. If you like that sort of thing, try getting the syntax slightly wrong (eg, omitting one of your variables from the "stubnames" list, and see what happens. You can always start again by using the SESSION1-2 data.
Use des and list to have a look at the data. There are now only four observations, but 47 variables. Note that there are 15 variables for age and LIKERT, but only one for sex and pid – and wave has vanished altogether (though we get it back when we convert back into long form)

Now, reshape the data back into long form.
reshape long LIKERT age, i(pid) j(wave)

des

Use list  to check that it's back to its old shape and size.
1.2.4
OLS USING A LAGGED DEPENDENT VARIABLE

Use your saved file SESSION1-2

use SESSION1-2, clear

We are going to re-run the OLS regression in section 1.7 using a lagged value of the dependent variable on the right-hand side.
We are going to re-estimate this including a lagged value of LIKERT on the right-hand side. First, we have to generate the lagged variable.

Create a variable LIKERT_LAG, as follows:
sort pid wave

gen LIKERT_LAG = LIKERT[_n-1] if pid == pid[_n-1] & wave == wave[_n-1] + 1

Check that you know what the command has done, and then inspect the data to check that the variable has been constructed properly:
sort pid wave

list pid wave LIKERT LIKERT_LAG in 1/200
Question: How many missing values are there for LIKERT and LIKERT_LAG? Why are there more missing values for LIKERT_LAG?

	


You may like to practice generating lagged values of other variables, or a lagged value of LIKERT which goes back two years. 
First, run the regression without the lagged dependent variable

reg LIKERT age age2 female ed_deg ed_sec partner ue_sick badhealth nch02, cluster(pid)
Now, run the regression including the lagged dependent variable. 
reg LIKERT LIKERT_LAG age age2 female ed_deg ed_sec partner ue_sick badhealth nch02, cluster(pid)
Compare the two sets of estimates. How do they differ?

	


1.2.5
MODELS OF CHANGE

These models regress the change in the dependent variable on the changes in the explanatory variables. There is always a better way to use the data, but let’s practise anyway. 

A bit of thought will tell us that age will not give us a very intelligent coefficient – it changes by 1 most years, and to the extent that it doesn’t, this has to do with interview dates.

Sex does not change from year to year for most people. Education levels may change occasionally – but let’s stick to three variables which are liable to change, and which we might think would have an effect on mental health: partner, ue_sick, and badhealth.

Generate a change variable for LIKERT (we’ll call it D_LIKERT because the Greek symbol delta Δ is used in mathematics to denote change)

gen D_LIKERT = LIKERT – LIKERT_LAG
Alternatively, we could have done it directly:

sort pid wave

gen D_LIKERT2 = LIKERT - LIKERT[_n-1] if pid == pid[_n-1] & wave == wave[_n-1] + 1

Now, in exactly the same way, generate change variables for the three explanatory variables: call them D_partner, D_ue_sick and D_badhealth. 

Finally, a change regression:

reg D_LIKERT D_partner D_ue_sick D_badhealth

What do these results tell us?
	


2.6
AT THE END OF THE SESSION

Save your working data set to your home directory, under the name SESSION1-2.dta
save SESSION1-2, replace






